PROM WRITER



User's Manual

ACKERMAN
DIGITAL SYSTEMS, INC.

Serial # 022274

ads

PROM BLASTER

S-100

USER'S MANUAL

Ackerman Digital Systems, Inc.
110 N. York Rd. - Suite 208
Elmhurst, Illinois 60126
(312) 530 - 8992

TABLE OF CONTENTS

I. Introduction..... 1

II. Board Construction..... 1

III. Using the ads PROMBLASTER..... 3

IV. Helpful Hints..... 6

V. Parts List..... 8

VI. Appendix - A sample software driver..... 9

VII. Schematic diagram.....15

WARNING

From time to time, unauthorized copies of ADS software such as the Promwriter or ADSMON, are sold over the counter or through distribution, the purchase of such software voids the warranty of both the software product and the hardware product it was purchased for. In its customer files, ADS keeps records of all product purchases, and any dealer/distributor violating software or hardware copyrights will be dropped.

ads PROMBLASTER User's Manual

I. Introduction

The ads PROMBLASTER is an S-100 compatible EPROM programming board. It appears to the computer as four I/O ports. The address, data and the amplitude and timing of the various programming pulses for each different EPROM size and family are controlled by software. Either 1k, 2k, 4k or 8k single or three supply, 24 or 28 pin EPROMS may be programmed. The PROMBLASTER has an on-card switching regulator to provide the high voltage for programming EPROMS. A one millisecond timing reference is also provided for controlling programming pulse widths. The ads PROMWRITER software provides full feature control of the PROMBLASTER and is available under CP/M.

II. Board Construction

1. Begin construction of the ads PROMBLASTER by first examining it for obvious shorts. If an ohmmeter is available measure between address lines, data lines, and the +5 volt and ground for shorts.

2. Noting their orientation against the silk screen, install and solder the I/C sockets. No socket should be used for the dip switches S1 and S2.

3. Carefully observing the polarized capacitors orientation against the silk screen, install and solder the capacitors.

4. Install the diodes next, matching their polarity with that of the silk screen.

5. The dip switches S1 and S2 and resistors should be installed next.

6. Now install the regulators and their heat sinks. Heat sink compound should be used sparingly. The regulators fasten to the heat sinks and board with 6-32 x 3/8 screws and nuts.

7. Install the inductor L1.

8. Install the transistors being careful to follow the emitter/collector/base pattern on the screen with the transistors you use. Note that the component designation appears next to the emitter pad.

9. Apply power to the board and verify that the output of the regulators match the values shown:

Q13 - +5vdc
Q14 - +5.7vdc
Q15 - +12.7vdc
Q16 - -5vdc

In addition, verify that +26vdc is available across C11.

10. Verify that +5vdc is present at the correct pins on the following I/C's:

I/C	Vcc	Gnd	Device
U1	20	10	74LS244
U2	20	10	74LS244
U3	20	10	74LS374
U4	20	10	74LS374
U5			There is no U5 part !
U6	20	10	74LS374
U7	20	10	74LS273
U8	28	14	Programming socket
U9	16	8	74LS155
U10	14	7	74LS27
U11	20	10	74LS682
U12	20	10	74LS682
U13	14	7	7406
U14	14	7	7406
U15	14	7	74LS02
U16	14	7	74LS27
U17	14	7	74LS04
U18	14	7	74LS02
U19	14	7	74LS08
U20	16	8	MC14020B
U21	14	7	74LS04
U22	14	7	74LS74
U23	14	7	74LS74
U24			There is no U24 part !
U25	16	8	74LS368
U26	14	5	TL497

11. Verify the following voltages on the pins of the programming socket U8:

U8 Pin	voltage
1	+5vdc
20	+5vdc
21	+5vdc
22	+5vdc
23	+5vdc
26	+5vdc

Now using a jumper wire, temporarily apply a ground to the

following points one at a time, and verify the U8 pin voltages:

Ground	- U8 Pin	- voltage
U13-6	- 1	- +26vdc
U13-4	- 20	- +26vdc
U13-8	- 21	- +12vdc
U13-2	- 22	- +26vdc
U14-2	- 22	- +12vdc
U13-12	- 23	- +26vdc
U14-6	- 23	- -5vdc
U14-12	- 26	- +12vdc

Again using a jumper wire, temporarily apply +5vdc to the following points one at a time, and verify the U8 pin voltages:

+5vdc	- U8 Pin	- voltage
U15-1	- 20	- 0vdc
U14-10	- 21	- 0vdc
U16-8	- 22	- 0vdc
U16-6	- 23	- 0vdc

12. Remove power and install the I/C's (do not bend over any pins and/or reverse the I/C's in their sockets).

13. The ads PROMBLASTER is designed to work in a 1 MHZ system. For use in faster systems the onboard wait state generator must be used. If you require one or two wait states for 2 or 4 MHZ systems install the necessary jumper (S,T,V - R) and select which of the two S-100 ready lines your system requires (N,P - M).

14. The ads PROMBLASTER may be used with standard or extended device addresses. For use in extended device address systems, use jumper D - H. If you are using the phantom slave option, connect the appropriate phantom polarity, E,F to jumper H. Otherwise jumper G - H should be installed.

15. The PROMBLASTER is now ready for use within your system. Select the group of four I/O addresses you want the board to respond to with switch S2. If you are using the extended device address option, you must also set switch S1 to the desired device page address.

16. Install the board and verify that your computer and other I/O devices function normally.

III. Using the ads PROMBLASTER

The ads PROMBLASTER is controlled through four I/O ports.

These are:

I/O Address	- Read Function	- Write Function
+ 00	- Prom data in	- Prom data out
+ 01	- Reset hi volt	- Prom A0-A7
+ 02	- Timer status	- Prom A8-A9, mode
+ 03	- Reset timer	- Prom hi volt control

In the following tables and descriptions the numbers P1-P28 refer to the pins on the device programming socket U8. Most EPROMs in a 24 or 28 pin package are functionally eqivalent on many of their pins as detailed below:



I/O address + 00 provides an eight bit data path to or from the programming socket U8. Data written to I/O address + 00 is latched and is presented to the EPROM data outputs when enabled. Reading from I/O address + 00 causes the data present on the EPROM data outputs to be input to the CPU.

I/O addr	B7	B6	B5	B4	B3	B2	B1	B0	
+00	P19	P18	P17	P16	P15	P13	P12	P11	READ/WRITE

I/O address + 01 provides a combined control and address function. Data written to I/O address + 01 is latched and is presented to the EPROM address lines A0 - A7 when enabled. Reading from I/O address + 01 causes the latch at I/O address + 03 to be cleared.

I/O addr	B7	B6	B5	B4	B3	B2	B1	B0	
+01	P3	P4	P5	P6	P7	P8	P9	P10	WRITE
+01	RESET I/O addr +03 latch contents								READ

I/O addresses + 02 and + 03 are combined timing, address and voltage control ports. Data written to I/O address + 02 enables/disables the data to the EPROM, controls the levels at three pins of U8 programming socket and supplies A8 - A9 as well. Data written to I/O address + 03 enables/disables the A0-A7 lines to the EPROM, and controls the levels at five pins of U8. Reading from I/O address + 02 causes the status of the one millisecond reference to be returned to the CPU on bit 7. Reading from I/O address + 03 causes the one millisecond timing reference to be restarted. To minimize the number of control bits needed, the voltages on the pins of U8 have been encoded such that two bits of I/O address + 02 set the mode of operation for many of the remaining bits. This is detailed below:

I/O addr +02 & +03 WRITE FUNCTIONS

I/O addr	B7	B6	B5	B4	B3	B2	B1	B0	P26	P23	P22	P21	P1	
	data		+5v	+5v		+5v								M
+02	out	1	P27	P2	1	P21	A9	A8						O
	dsbl		0v	0v		0v								D
	addr	+5v	+5v	+5v	+25v	+25v	+25v	1	+5v				+25v	E
+03	out	P23	P22	P20	P23	P22	P20							3
	enbl	0v	0v	0v	B6	B5	B4	0	+5v				+5v	
	data		+5v	+5v										M
+02	out	1	P27	P2	0	na	A9	A8						O
	dsbl		0v	0v										D
	addr	+5v	+5v	+5v	+25v	+25v	+25v	1	+12v	-5v		+12v	+25v	E
+03	out	P23	P22	P20	P23	P22	P20							2
	enbl	0v	0v	0v	B6	B5	B4	0	+5v	-5v		+12v	+5v	
	data		+5v	+5v		+5v								M
+02	out	0	P27	P2	1	P21	A9	A8						O
	dsbl		0v	0v		0v								D
	addr	+5v	+5v	+5v	+25v	+25v	+25v	1	+5v		+12v		+25v	E
+03	out	P23	P22	P20	P23	P22	P20							1
	enbl	0v	0v	0v	B6	B5	B4	0	+5v		B5		+5v	
	data		+5v	+5v										M
+02	out	0	P27	P2	0	na	A9	A8						O
	dsbl		0v	0v										D
	addr		+5v	+5v	+25v	+25v	+25v	1	+5v	-5v	+12v	+12v	+25v	E
+03	out	na	P22	P20	P23	P22	P20							0
	enbl		0v	0v	-5v	B5	B4	0	+5v	-5v	B5	+12v	+5v	

I/O addr +02 & +03 READ FUNCTIONS

I/O addr	B7	B6	B5	B4	B3	B2	B1	B0
+02	1 msec timer status							
+03	Reset 1 millisecond timer							

Before an EPROM can be put into U8 the socket must be setup with the correct voltage configuration, the data outputs must be disabled, the address inputs should be disabled, and the chip select pin for the part should be made inactive. This is accomplished be loading the correct bit patterns into I/O addresses + 02 and +03. After placing the correct EPROM into U8, it may be read by latching the desired address into A0 - A7 via I/O address +01, setting up A8 - A?? via I/O addresses +02 and +03 and enabling the address to the EPROM. This is followed by making the desired chip select and/or output enable line active, again via I/O addresses +02 and +03, and then reading the data from the EPROM via I/O address +00.

Programming an EPROM involves more steps. After placing the EPROM into a properly configured U8 socket, the desired address is setup as for reading above. The EPROM must now be placed into programming mode. For single supply parts this may be as simple as raising the Vpp pin to +25v. Three supply parts require an additional pin be brought to +12v. The data to be programmed is latched into I/O address +00 and is enabled to the U8 socket. A programming pulse varying from one to fify milliseconds is then supplied to the EPROM. After this pulse, the data is disabled, a new address is supplied, and the process repeats. Single supply EPROMS may be programmed at any location randomly. Three supply parts require a continuous looping through all locations with null data supplied to the unaffected locations. A software example for Intel 2716 +5v EPROMS may be found in the Appendix.

IV. Helpful Hints

After writing PROMBLASTER software, check out its operation with a scope or multimeter first. Observe the relationship, amplitude and timing of the signals present on the programming socket U8 before attempting to read or program your part. It is very easy to toggle the incorrect bit causing the PROMBLASTER to apply the wrong voltage to

the wrong pin. This may result in damage to the EPROM and/or PROMBLASTER. Mode 3 is used for all single supply EPROMS. Mode 2 is provided for three supply 2K parts. Mode 1 is not used. Mode 0 is for three supply 1K EPROMS. Note that through proper control it should be possible to read 24 and 28 pin ROMS as well, provided that you know the mask-programmed chip select levels.

The ads PROMWRITER software is a package written in 8080 assembly language for execution under CP/M. It provides the capability of programming 14 different EPROMS, both single or three supply parts, 1K, 2K, 4K and 8K parts, both 24 and 28 pin packages when used with the ads PROMBLASTER. Commands are provided to read, program and verify EPROMS with optional offsets. Memory load, examine, display and sum functions as well as CP/M hex file load operations are also provided.

WARNING

From time to time, unauthorized copies of ADS software such as the Promwriter or Adsmon, are sold over the counter or through distribution, the purchase of such software voids the warranty of both the software product and the hardware product it was purchased for. In its customer files, ADS keeps records of all product purchases, and any dealer/distributor violating software or hardware copyrights will be dropped.

VI. Parts List

Quantity	Part Number	Description
2	U15, U18	74LS02 QUAD 2 IN Nor
2	U17, U21	74LS04 Hex Inverter
2	U13-14	7406 Hex In erter OC HV
1	U19	74LS08 Quad 2 In and
2	U10, U16	74LS27 Triple 3 In Nor
2	U22-23	74LS74 Dual D Flip Flop
1	U9	74LS155 Dual 2 to 4 Decoder
2	U1-2	74LS244 Octal Bus Buffer
1	U7	74LS273 Octal D Flip Flop w/Clr
1	U25	74LS368 Hex Bus Inverter W/3 S
3	U3, U4, U6	74LS374 ~~74LS375~~ Octal D Flip Flop W/3 S
2	U11-12	74LS682 Octal Comparator W/Pulls
1	U20	MC14020B 14 Stage Binary Counter
1	U26	TL497 Switching Regulator
2	Q13-14	LM340T-5 5V to-220 Regulator
1	Q15	LM340T-12 12V to-220 Regulator
1	Q16	LM320T-5 5V to-220 Regulator
4	Q9-12	2N2222 NPN GP Transistor
8	Q1-8	2N4403 PNP GP Transistor
15	CR1-CR15	1N4001 ~~1N4148~~ 1A 50 PIV Diode
1	R69	2 OHM 1/2 Watt 5% Resistor
2	R66-67	150 OHM 1/4 Watt 5% Resistor
1	R63	150 OHM 1 Watt 5% Resistor
1	R20	750 OHM 1/4 Watt 5% Resistor
1	R71	1.2K OHM 1/4 Watt 5% Resistor
13	R2,R4,R6,R8-R10 R12-16, R64-65	2.2K OHM 1/4 Watt 5% Resistor
5	R40, R50-52,R62	3.0K OHM 1/4 Watt 5% Resistor
4	R1,R3,R5,R7	3.3K OHM 1/4 Watt 5% Resistor
3	R17-19	5.1K OHM 1/4 Watt 5% Resistor
2	R11,R68	10K OHM 1/4 Watt 5% Resistor
1	R70	24.9K ~~24K~~ OHM 1/4 Watt 5% Resistor
1	C17	10 PFD Ceramic Disc Capacitor
1	C13	330 PFD Ceramic Disc Capacitor'
1	C16	.002 UFD Ceramic Disc Capacitor
23	C3,C4,C7,C10,C12 C14 & C20-C37	.01 UFD Ceramic Disc Capacitor
5	C1-2,C8-9,C15	4.7 UFD 25V Tantalum Capacitor
5	C5-6,C11,C18-19	10 UFD 35V Aluminum Capacitor
1	L1	250 UH 0.2A Inductor
2	S1,S2	8 Position Dip Switch
2		THM6106 to-220 Heatsink
2		THM6073 To-220 Heatsink
8	U1-2,U3-4,U6-7,U11-12	20 Pin I.C. Socket
3	U9,U20,U25	16 Pin I.C. Socket
12	U10,U13-16 U17-19,U21-23,U26	14 Pin I.C. Socket
1	U8	28 Pin L.I.F./Z.I.F.I.C. Socket

NOTE: There is no U5 or U24

VII. Appendix - a software example

The following is a 8080 code example for the ads PROMBLASTER. It allows programming, verifying, and reading of INTEL 2716 EPROMS. After assembly and loading, it is invoked via DDT. After execution at the various entry points in the function table, control is returned to DDT with a RST 07 instruction. The result of the function is returned in the Z flag. A non-zero Z flag indicates successful completion of the function jumped to. Note that the PROMBLASTER I/O routines maintain a RAM copy of the I/O port's status to allow setting and resetting of individual bits. This code segment is presented as an example of PROMBLASTER control software. Full feature PROMBLASTER control is available with the ads PROMWRITER software.

```
                        ;
                        ; ADS PROMBLASTER INTEL 2716 EXAMPLE
                        ;
00C0 =          PRMBAS  EQU     0C0H    ;PROMBLASTER BASE PORT #
                        ;
0100                    ORG     0100H
                        ;
                        ; FUNCTION JUMP TABLE
                        ;
0100 CD6701             CALL    CONFIG  ;CONFIGURE PROMBLASTER FOR I2716
0103 FF                 RST     7       ;CALL DDT
0104 CDDA01             CALL    CHECK   ;CHECK I2716 FOR UNBURNED STATE
0107 FF                 RST     7       ;CALL DDT
0108 CD6A01             CALL    PROGRM  ;PROGRAM I2716 FROM -> TO
010B FF                 RST     7       ;CALL DDT
010C CD8F01             CALL    VERIFY  ;VERIFY I2716 FROM -> TO
010F FF                 RST     7       ;CALL DDT
0110 CDB501             CALL    READ    ;READ I2716 FROM -> TO
0113 FF                 RST     7       ;CALL DDT
0114 C30000             JMP     0       ;CALL CP/M
                        ;
                        ; RAM DEFINITIONS
                        ;
0117 0000       FROM:   DW      0       ;FROM LOCATION
0119 0000       TO:     DW      0       ;TO LOCATION
011B 0000       POFF:   DW      0       ;PROM OFFSET
011D 00         PRDT:   DB      0       ;PROMBLASTER DATA PORT SAVE
011E 00         ADLO:   DB      0       ;PROMBLASTER ADDR LOW SAVE
011F 00         ADHI:   DB      0       ;PROMBLASTER ADDR HI SAVE
0120 00         VCTL:   DB      0       ;PROMBLASTER VOLTAGE CONTROL SAVE
                        ;
                        ; PROMBLASTER I/O ROUTINES
                        ;
0121 DBC0       PRDTIN: IN      PRMBAS
0123 C9                 RET
                        ;
0124 D3C0       PRDTOT: OUT     PRMBAS
0126 C9                 RET
```

```
                        ;
0127 321E01     ADLOOT:STA      ADLO
012A D3C1               OUT     PRMBAS+1
012C C9                 RET
                        ;
012D 47         ORADHI:MOV      B,A
012E 3A1F01             LDA     ADHI
0131 B0                 ORA     B
0132 321F01     ADHIOT:STA      ADHI
0135 D3C2               OUT     PRMBAS+2
0137 C9                 RET
                        ;
0138 47         ANADHI:MOV      B,A
0139 3A1F01             LDA     ADHI
013C A0                 ANA     B
013D C33201             JMP     ADHIOT
                        ;
0140 DBC1       RSVCTL:IN       PRMBAS+1
0142 C9                 RET
                        ;
0143 47         ORVCTL:MOV      B,A
0144 3A2001             LDA     VCTL
0147 B0                 ORA     B
0148 322001     VCTLOT:STA      VCTL
014B D3C3               OUT     PRMBAS+3
014D C9                 RET
                        ;
014E 47         ANVCTL:MOV      B,A
014F 3A2001             LDA     VCTL
0152 A0                 ANA     B
0153 C34801             JMP     VCTLOT
                        ;
0156 CD6001     WAITIM:CALL     RSTTIM
0159 CD6301     WAITLP:CALL     CHKTIM
015C D25901             JNC     WAITLP
015F C9                 RET
                        ;
0160 DBC3       RSTTIM:IN       PRMBAS+3
0162 C9                 RET
                        ;
0163 DBC2       CHKTIM:IN       PRMBAS+2
0165 17                 RAL
0166 C9                 RET
                        ;
                        ; CONFIGURE THE PROGRAMMING SOCKET - U8
                        ;
0167 =          CONFIG EQU      $
0167 C30002             JMP     I2716C  ;INTEL 2716
                        ;
                        ; PROGRAM I2716 FOR FROM -> TO LOCATIONS
                        ;
016A =          PROGRM EQU      $
016A 210010             LXI     H,1000H ;FROM=1000H
016D 221701             SHLD    FROM    ;*
0170 21FF17             LXI     H,17FFH ;TO=FROM+2048
```

WARNING

From time to time, unauthorized copies of ADS software such as the Promwriter or Adsmon, are sold over the counter or throug distribution, the purchase of such software voids the warranty of both the software product and the hardware product it was purchased for. In its customer files, ADS keeps records of all product purchases, and any dealer/ distributor violating software or hardware copyrights will be dropped.

```
0173 221901            SHLD    TO        ;*
0176 =          PROGLP EQU     $
0176 CD2802            CALL    I2716P    ;PROGRAM @ FROM
0179 2A1901            LHLD    TO        ;Q-FROM = TO?
017C EB                XCHG              ;*
017D 2A1701            LHLD    FROM      ;*
0180 7C                MOV     A,H       ;*
0181 BA                CMP     D         ;*
0182 C28801            JNZ     NXTP      ;*
0185 7D                MOV     A,L       ;*
0186 BB                CMP     E         ;*
0187 C8                RZ                ;RETURN IF FROM = TO
0188 =          NXTP   EQU     $
0188 23                INX     H         ;FROM = FROM +1
0189 221701            SHLD    FROM      ;*
018C C37601            JMP     PROGLP    ;CONTINUE PROGRAMMING
                ;
                ; VERIFY I2716 FOR FROM -> TO LOCATIONS
                ;
018F =          VERIFY EQU     $
018F 210010            LXI     H,1000H           ;SETUP FROM
0192 221701            SHLD    FROM              ;*
0195 21FF17            LXI     H,17FFH           ;SETUP TO = FROM + 2048
0198 221901            SHLD    TO                ;*
019B =          VERILP EQU     $
019B CD8002            CALL    I2716V            ;VERIFY I2716 @ FROM
019E C0                RNZ                       ;ERROR @ FROM
019F 2A1901            LHLD    TO                ;ADVANCE FROM, FROM=TO?
01A2 EB                XCHG                      ;*
01A3 2A1701            LHLD    FROM              ;*
01A6 7C                MOV     A,H               ;*
01A7 BA                CMP     D                 ;*
01A8 C2AE01            JNZ     NXTV              ;*
01AB 7D                MOV     A,L               ;*
01AC BB                CMP     E                 ;*
01AD C8                RZ                        ;RETURN IF DONE
01AE =          NXTV   EQU     $
01AE 23                INX     H                 ;FROM = FROM +1
01AF 221701            SHLD    FROM              ;*
01B2 C39B01            JMP     VERILP            ;CONTINUE VERIFYING
                ;
                ; READ I2716 FOR FROM -> TO LOCATIONS
                ;
01B5 =          READ   EQU     $
01B5 210010            LXI     H,1000H           ;SETUP FROM
01B8 221701            SHLD    FROM              ;*
01BB 21FF17            LXI     H,17FFH           ;SETUP TO = FROM +2048
01BE 221901            SHLD    TO                ;*
01C1 =          READLP EQU     $
01C1 CD8902            CALL    I2716R            ;READ I2716 @ FROM
01C4 2A1901            LHLD    TO                ;ADVANCE FROM, FROM=TO?
01C7 EB                XCHG                      ;*
01C8 2A1701            LHLD    FROM              ;*
01CB 7C                MOV     A,H               ;*
01CC BA                CMP     D                 ;*
```

```
01CD C2D301             JNZ     NXTR            ;*
01D0 7D                 MOV     A,L             ;*
01D1 BB                 CMP     E               ;*
01D2 C8                 RZ                      ;RETURN IF DONE
01D3 =          NXTR    EQU     $
01D3 23                 INX     H               ;FROM=FROM + 1
01D4 221701             SHLD    FROM            ;*
01D7 C3C101             JMP     READLP          ;CONTINUE READING
                ;
                ; CHECK I2716 FOR FROM -> TO UNBURNED LOCATIONS
                ;
01DA =          CHECK   EQU     $
01DA 210010             LXI     H,1000H         ;SETUP FROM
01DD 221701             SHLD    FROM            ;*
01E0 21FF17             LXI     H,17FFH         ;SETUP TO
01E3 221901             SHLD    TO              ;*
01E6 =          CHEKLP  EQU     $
01E6 CD9102             CALL    I2716U          ;CHECK I2716 @ FROM FOR UNBURN
01E9 C0                 RNZ                     ;RETURN IF NOT
01EA 2A1901             LHLD    TO              ;ADVANCE FROM, FROM = TO?
01ED EB                 XCHG                    ;*
01EE 2A1701             LHLD    FROM            ;*
01F1 7C                 MOV     A,H             ;*
01F2 BA                 CMP     D               ;*
01F3 C2F901             JNZ     NXTC            ;*
01F6 7D                 MOV     A,L             ;*
01F7 BB                 CMP     E               ;*
01F8 C8                 RZ                      ;RETURN IF DONE
01F9 =          NXTC    EQU     $
01F9 23                 INX     H               ;FROM =FROM +1
01FA 221701             SHLD    FROM            ;*
01FD C3E601             JMP     CHEKLP          ;CONTINUE CHECKING
                ;
                ; SINGLE SUPPLY 2K PARTS
                ;
                ;       INT2716,MCM2716,TMS2516
                ;---------------------------------------
                ;
                ; CONFIGURE PROMBLASTER U8 SOCKET FOR I2716
                ;
0200 =          I2716C  EQU     $
0200 3E60               MVI     A,01100000B     ;VPP=5V,G=5V
0202 CD4801             CALL    VCTLOT          ;*
0205 3EC8               MVI     A,11001000B     ;DISBL DATA, MODE 3
0207 CD3201             CALL    ADHIOT          ;*
020A AF                 XRA     A               ;ADDR LO=0, DATA = 0
020B CD2701             CALL    ADLOOT          ;*
020E CD2401             CALL    PRDTOT          ;*
0211 C9                 RET
                ;
                ; SETUP A0-A10 SUBROUTINE
                ;
0212 =          I2716S  EQU     $
0212 3A1701             LDA     FROM            ;ADDRESS SETUP
0215 CD2701             CALL    ADLOOT          ;*A0-A7
```

```
0218 3A1F01             LDA     ADHI            ;*A8-A10
021B E6F8               ANI     11111000B       ;**
021D 47                 MOV     B,A             ;**
021E 3A1801             LDA     FROM+1          ;**
0221 E607               ANI     00000111B       ;**
0223 B0                 ORA     B               ;**
0224 CD3201             CALL    ADHIOT          ;**
0227 C9                 RET
                ;
                ; PROGRAM I2716 @ FROM SUBROUTINE
                ;
0228 =          I2716P  EQU     $
0228 CD1202             CALL    I2716S          ;ADDRESS SETUP
022B 3E80               MVI     A,10000000B     ;ADDR ENBL
022D CD4301             CALL    ORVCTL          ;*
0230 3E08               MVI     A,00001000B     ;VPP=25V
0232 CD4301             CALL    ORVCTL          ;*
0235 2A1701             LHLD    FROM            ;SETUP DATA
0238 7E                 MOV     A,M             ;*
0239 CD2401             CALL    PRDTOT          ;*
023C 3E7F               MVI     A,01111111B     ;ENBL DATA
023E CD3801             CALL    ANADHI          ;*
0241 3E10               MVI     A,00010000B     ;PROGR=5V
0243 CD4301             CALL    ORVCTL          ;*
0246 0632               MVI     B,50            ;WAIT FOR 50 MSEC
0248 =          I2716L  EQU     $               ;*
0248 CD5601             CALL    WAITIM          ;*
024B 05                 DCR     B               ;*
024C C24802             JNZ     I2716L          ;*
024F 3EEF               MVI     A,11101111B     ;PROGR=0V
0251 CD4E01             CALL    ANVCTL          ;*ROGR=0V
0254 3EF7               MVI     A,11110111B     ;VPP=5V
0256 CD4E01             CALL    ANVCTL          ;*
0259 3E80               MVI     A,10000000B     ;DISBL DATA
025B CD2D01             CALL    ORADHI          ;*
025E 3E7F               MVI     A,01111111B     ;DISBL ADDR
0260 CD4E01             CALL    ANVCTL          ;*
0263 C9                 RET
                ;
                ; READ I2716 @ FROM SUBROUTINE
                ;
0264 =          I2716Q  EQU     $
0264 CD1202             CALL    I2716S          ;ADDR SETUP
0267 3E80               MVI     A,10000000B     ;ENBL ADDR
0269 CD4301             CALL    ORVCTL          ;*
026C 3EDF               MVI     A,11011111B     ;G=0V
026E CD4E01             CALL    ANVCTL          ;*
0271 CD2101             CALL    PRDTIN          ;READ PROM
0274 4F                 MOV     C,A             ;SAVE
0275 3E20               MVI     A,00100000B     ;G=5V
0277 CD4301             CALL    ORVCTL          ;*
027A 3E7F               MVI     A,01111111B     ;DISBL ADDR
027C CD4E01             CALL    ANVCTL          ;*
027F C9                 RET
                ;
```

```
                ; VERIFY I2716 @ FROM SUBROUTINE
                ;
0280 =          I2716V EQU      $
0280 CD6402            CALL     I2716Q          ;READ PROM
0283 2A1701            LHLD     FROM            ;PNT2 DATA
0286 7E                MOV      A,M             ;GET DATA
0287 B9                CMP      C               ;TEST AGAINST PROM
0288 C9                RET
                ;
                ; READ I2716 @ FROM SUBROUTINE
                ;
0289 =          I2716R EQU      $
0289 CD6402            CALL     I2716Q          ;READ PROM
028C 2A1701            LHLD     FROM            ;PNT2 TARGET ADDR
028F 71                MOV      M,C             ;SAVE DATA READ
0290 C9                RET
                ;
                ; CHECK I2716 @ FROM FOR UNBURNED SUBROUTINE
                ;
0291 =          I2716U EQU      $
0291 CD6402            CALL     I2716Q          ;READ PROM
0294 3EFF              MVI      A,11111111B
0296 B9                CMP      C               ;TEST PROM FOR UNBURNED
0297 C9                RET
                ;
0298                   END
```

WARRANTY CARD

Please fill this out and mail to us to activate your 90 days warranty. By returning this card, you will also be on our mailing list for application notes, error corrections, and new product announcements.

NAME _____

ADDRESS _____

CITY _____

STATE _____ ZIP _____

NAME OF COMPANY _____
or distributor you
purchased the Promblaster _____
from

Did you also purchase the Promwriter _____ If so, was it from the above named company or another?

_____ which version? _____

Computer System _____

Application _____

How did you learn of our products? _____

What future products would you like to see? _____

THANK YOU.

Ackerman Digital Systems, Inc.
110 No. York, Suite 208
Elmhurst, Il. 60126

JUN 2 1982

DATE

APPROVED BY

POC* 38 — U19 LS08

SLAVE CLR* 54

S1

A15 31 A14 32 A13 33 A12 34 A11 35 A10 37 A9 36 A8 39

U12 LS682 LS683

P=Q 19

+5V — R77

S2

A7 83 A6 82 A5 23 A4 30 A3 31 A2 81

U11 LS682 LS683

U18 LS02 — 46 45

P=Q 19

+5V — R78

A1 80 A0 79

U9 LS155

2G 15 — 1C 1 — 2C 15

pDBIN 78

PWR* 77

U10 LS27

OP1 EXTENDED DEVICE OPTION

pSTVAL* 25

OP2 CLOCK OPTION

pSYNC 76

OP3 QUALIFIER OPTION

U16 LS27

Φ 24

U23 LS74

WAIT STATES

PRDY 72

XRDY 3

CLOCK 58

U4 LS374

U3 LS374

U8 L.I.F. SOCKET

Vpp Vcc Vss 1K

+5V — R62

U19 LS08

Q2 CR12 R4 R3 R19 Q11 CR13 R67 +5.7V

U15 LS02

U22 MC14020B

RST

C16 RCB +5V

U18 LS02

HV T.P. +HV

C11 C12 R70 R71

U26 TL497

C13 C14 C15

Q14 R74 R75 R76

DN TOP2 TO92 2N2222

CR3 CR4 CR5 CR6 Q3 Q8 R6 R16 R5 R51 +5.7V +HV

U19 LS08 U10 LS27 U15 LS02

CR7 CR8 Q4 Q9 R8 R7 R17 R63 R65 R64 -5V

Q12 Q6 R12 R11

CR9 CR10 CR11 CR1 CR2 Q1 Q10 Q7 Q5 Q13 R2 R1 R18 R14 R13 R66 R10 R9 R73 R20 +12.7V

U18 LS02 R52 +5V

U4 LS374 U6 LS374 U7 LS273

U1 LS244 U2 LS244

DO0 DO1 DO2 DO3 DO4 DO5 DO6 DO7

DI7 DI6 DI5 DI4 DI3 DI2 DI1 DI0

C17 R40 +5V

OPT. GND 53 GND

THM6106 Q13 LM340T5 +8V C7 C8 C19

THM6106 Q14 LM340T5 C9 C10

CR14 OUT GRD INPUT

THM6073 Q15 LM340T12 +16V C3 C5

CR15 THM6073 Q16 LM320MP5 -16V C4 C6

5V T.P. 12.7V T.P. -5 T.P.

C1 C2

PINT 73 pINT*

NINI 12 NINI*

check 26 volts

ads PROMBLASTER

SCALE: N/A	APPROVED BY:	DRAWN BY SJA
DATE: 5/18/81		REVISED 4/18/82
		DRAWING NUMBER PB-2

17X22 PRINTED ON NO. 1000H-S CLEARPRINT FADE-OUT